NAG Fortran Library Routine Document

F08JVF (ZSTEDC)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08JVF (ZSTEDC) computes all the eigenvalues and, optionally, all the eigenvectors of a real n by n symmetric tridiagonal matrix, or of a complex full or banded Hermitian matrix which has been reduced to tridiagonal form.

2 Specification

```
 SUBROUTINE F08JVF (COMPZ, N, D, E, Z, LDZ, WORK, LWORK, RWORK, LRWORK,
1                   IWORK, LIWORK, INFO)

 INTEGER             N, LDZ, LWORK, LRWORK, IWORK(*), LIWORK, INFO
 double precision    D(*), E(*), RWORK(*)
 complex*16          Z(LDZ,*), WORK(*)
 CHARACTER*1         COMPZ
```

The routine may be called by its LAPACK name ***zstedc***.

3 Description

F08JVF (ZSTEDC) computes all the eigenvalues, and optionally the eigenvectors, of a real symmetric tridiagonal matrix T. That is, the routine computes the spectral factorization of T given by

$$T = Z\Lambda Z^{\mathrm{T}},$$

where Λ is a diagonal matrix whose diagonal elements are the eigenvalues, λ_i, of T and Z is an orthogonal matrix whose columns are the eigenvectors, z_i, of T. Thus

$$Tz_i = \lambda_i z_i, \quad i = 1, 2, \ldots, n.$$

The routine may also be used to compute all the eigenvalues and vectors of a real full, or banded, Hermitian matrix A which has been reduced to real tridiagonal form T as

$$A = QTQ^{\mathrm{H}},$$

where Q is unitary. The spectral factorization of A is then given by

$$A = (QZ)\Lambda(QZ)^{\mathrm{H}}.$$

In this case Q must be formed explicitly and passed to F08JVF (ZSTEDC) in the array Z, and the routine called with COMPZ = 'V'. Routines which may be called to form T and Q are

full matrix	F08FSF (ZHETRD) and F08FTF (ZUNGTR)
full matrix, packed storage	F08GSF (ZHPTRD) and F08GTF (ZUPGTR)
band matrix	F08HSF (ZHBTRD), with VECT = 'V'

When only eigenvalues are required then this routine calls F08JFF (DSTERF) to compute the eigenvalues of the tridiagonal matrix T, but when eigenvectors of T are also required and the matrix is not too small, then a divide and conquer method is used, which can be much faster than F08JSF (ZSTEQR), although more storage is required.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: COMPZ – CHARACTER*1 *Input*

On entry: indicates whether the eigenvectors are to be computed.

COMPZ = 'N'

Only the eigenvalues are computed (and the array Z is not referenced).

COMPZ = 'I'

The eigenvalues and eigenvectors of T are computed (and the array Z is initialized by the routine).

COMPZ = 'V'

The eigenvalues and eigenvectors of A are computed (and the array Z must contain the matrix Q on entry).

Constraint: COMPZ = 'N', 'V' or 'I'.

2: N – INTEGER *Input*

On entry: n, the order of the symmetric tridiagonal matrix T.

Constraint: $\mathrm{N} \geq 0$.

3: D(∗) – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: the diagonal elements of the tridiagonal matrix.

On exit: if INFO = 0, the eigenvalues in ascending order.

4: E(∗) – ***double precision*** array *Input/Output*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N}-1)$.

On entry: the subdiagonal elements of the tridiagonal matrix.

On exit: the array is overwritten.

5: Z(LDZ,∗) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$.

On entry: if COMPZ = 'V', Z must contain the unitary matrix used in the reduction to tridiagonal form.

On exit: if INFO = 0, then if COMPZ = 'V', Z contains the orthonormal eigenvectors of the original Hermitian matrix,and if COMPZ = 'I', Z contains the orthonormal eigenvectors of the symmetric tridiagonal matrix.

If COMPZ = 'N', Z is not referenced.

6: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

Constraints:

if COMPZ = 'V' or 'I', $\mathrm{LDZ} \geq \max(1, \mathrm{N})$;
$\mathrm{LDZ} \geq 1$ otherwise.

7: WORK(∗) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least max(1, LWORK).

On exit: if INFO = 0, WORK(1) returns the minimum LWORK.

8: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

If LWORK = −1, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK, RWORK and IWORK arrays, returns these values as the first entries of the WORK, RWORK and IWORK arrays, and no error message related to LWORK, LRWORK or LIWORK is issued.

Constraints:

if LWORK ≠ −1,
if COMPZ = 'N' or 'I' or N ≤ 1, LWORK must be at least 1;
if COMPZ = 'V' and N > 1, LWORK must be at least N^2.

Note: that for COMPZ = 'V', then if N is less than or equal to the minimum divide size, usually 25, then LWORK need only be 1.

9: RWORK(∗) – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least max(1, LRWORK).

On exit: if INFO = 0, RWORK(1) returns the optimal LRWORK.

10: LRWORK – INTEGER *Input*

On entry: the dimension of the array RWORK as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

If LRWORK = −1, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK, RWORK and IWORK arrays, returns these values as the first entries of the WORK, RWORK and IWORK arrays, and no error message related to LWORK, LRWORK or LIWORK is issued.

Constraints:

if LRWORK ≠ −1,
if COMPZ = 'N' or N ≤ 1, LRWORK must be at least 1;
if COMPZ = 'V' and N > 1,
LRWORK must be at least $1 + 3 \times \mathrm{N} + 2 \times \mathrm{N} \times lg(\mathrm{N}) + 3 \times \mathrm{N}^2$, where $lg(\mathrm{N})$ = smallest integer k such that $2^k \geq \mathrm{N}$;
if COMPZ = 'I' and N > 1, LRWORK must be at least $1 + 4 \times \mathrm{N} + 2 \times \mathrm{N}^2$.

Note: that for COMPZ = 'I' or 'V' then if N is less than or equal to the minimum divide size, usually 25, then LRWORK need only be max(1, 2 × (N − 1)).

11: IWORK(∗) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least max(1, LIWORK).

On exit: if INFO = 0, IWORK(1) returns the optimal LIWORK.

12: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

If LIWORK = −1, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK, RWORK and IWORK arrays, returns these values as the first entries of the WORK,

RWORK and IWORK arrays, and no error message related to LWORK, LRWORK or LIWORK is issued.

Constraints:

if LIWORK $\neq -1$,
if COMPZ = 'N' or N ≤ 1, LIWORK must be at least 1;
if COMPZ = 'V' or N > 1, LIWORK must be at least $6 + 6 \times \mathrm{N} + 5 \times \mathrm{N} \times lg(\mathrm{N})$;
if COMPZ = 'I' or N > 1, LIWORK must be at least $3 + 5 \times \mathrm{N}$.

13: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm failed to compute an eigenvalue while working on the sub-matrix lying in rows and columns INFO$/(\mathrm{N}+1)$ through mod(INFO, N + 1).

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(T + E)$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the ***machine precision***.

If λ_i is an exact eigenvalue and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$\left|\tilde{\lambda}_i - \lambda_i\right| \leq c(n)\epsilon\|T\|_2,$$

where $c(n)$ is a modestly increasing function of n.

If z_i is the corresponding exact eigenvector, and $\tilde{z}_i$ is the corresponding computed eigenvector, then the angle $\theta(\tilde{z}_i, z_i)$ between them is bounded as follows:

$$\theta(\tilde{z}_i, z_i) \leq \frac{c(n)\epsilon\|T\|_2}{\min\limits_{i \neq j}\left|\lambda_i - \lambda_j\right|}.$$

Thus the accuracy of a computed eigenvector depends on the gap between its eigenvalue and all the other eigenvalues.

See Anderson *et al.* (1999) (Section 4.7) for further details. See also F08FLF (DDISNA).

8 Further Comments

If only eigenvalues are required, the total number of floating point operations is approximately proportional to n^2. When eigenvectors are required the number of operations is bounded above by approximately the same number of operations as F08JSF (ZSTEQR), but for large matrices F08JVF (ZSTEDC) is usually much faster.

The real analogue of this routine is F08JHF (DSTEDC).

9 Example

This example finds all the eigenvalues and eigenvectors of the Hermitian band matrix

$$A = \begin{pmatrix} -3.13 & 1.94 - 2.10i & -3.40 + 0.25i & 0 \\ 1.94 + 2.10i & -1.91 & -0.82 - 0.89i & -0.67 + 0.34i \\ -3.40 - 0.25i & -0.82 + 0.89i & -2.87 & -2.10 - 0.16i \\ 0 & -0.67 - 0.34i & -2.10 + 0.16i & 0.50 \end{pmatrix}.$$

A is first reduced to tridiagonal form by a call to F08HSF (ZHBTRD).

9.1 Program Text

```
*     F08JVF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          LGNMAX, NMAX, KDMAX
      PARAMETER        (LGNMAX=5,NMAX=2**LGNMAX,KDMAX=8)
      INTEGER          LDAB, LDQ, LIWORK, LRWORK, LWORK
      PARAMETER        (LDAB=KDMAX+1,LDQ=NMAX,
     +                 LIWORK=6+6*NMAX+5*NMAX*LGNMAX,
     +                 LRWORK=1+3*NMAX+2*NMAX*LGNMAX+3*NMAX*NMAX,
     +                 LWORK=NMAX*NMAX)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, LIWOPT, LRWOPT, LWOPT, N
*     .. Local Arrays ..
      COMPLEX *16      AB(LDAB,NMAX), Q(LDQ,NMAX), WORK(LWORK)
      DOUBLE PRECISION D(NMAX), E(NMAX-1), RWORK(LRWORK)
      INTEGER          IWORK(LIWORK)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         X04DBF, ZHBTRD, ZSTEDC
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08JVF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
*        Read the upper or lower triangular part of the band matrix A
*        from data file
*
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
*
*        Reduce A to tridiagonal form T = (Q**T)*A*Q, and form Q
*
         CALL ZHBTRD('V',UPLO,N,KD,AB,LDAB,D,E,Q,LDQ,WORK,INFO)
*
*        Calculate all the eigenvalues and eigenvectors of A,
*        from T and Q
*
         CALL ZSTEDC('V',N,D,E,Q,LDQ,WORK,LWORK,RWORK,LRWORK,IWORK,
     +               LIWORK,INFO)
         LWOPT = WORK(1)
```

```
         LRWOPT = RWORK(1)
         LIWOPT = IWORK(1)
*
         IF (INFO.EQ.0) THEN
*
*           Print eigenvalues and eigenvectors
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (D(I),I=1,N)
*
            WRITE (NOUT,*)
            IFAIL = 0
            CALL X04DBF('General',' ',N,N,Q,LDQ,'Bracketed','F7.4',
     +                  'Eigenvectors','Integer',RLABS,'Integer',CLABS,
     +                  80,0,IFAIL)
*
         ELSE
            WRITE (NOUT,99998) 'Failure in ZSTEDC. INFO = ', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99997) 'Complex workspace required = ', LWOPT,
     +        'Complex workspace provided = ', LWORK
         END IF
         IF (LRWORK.LT.LRWOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99997) 'Real workspace required    = ', LRWOPT,
     +        'Real workspace provided    = ', LRWORK
         END IF
         IF (LIWORK.LT.LIWOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99997) 'Integer workspace required = ', LIWOPT,
     +        'Integer workspace provided = ', LIWORK
         END IF
       ELSE
          WRITE (NOUT,*) 'NMAX and/or KDMAX too small'
       END IF
       STOP
*
99999 FORMAT (4X,F8.4,3(10X,F8.4))
99998 FORMAT (1X,A,I10)
99997 FORMAT ((1X,A,I5))
      END
```

9.2 Program Data

```
F08JVF Example Program Data

    4              2                                        :Values of N and KD

 ( -3.13 , 0.00) (  1.94, -2.10) ( -3.40,  0.25)
                 ( -1.91,  0.00) ( -0.82, -0.89) ( -0.67,  0.34)
                                 ( -2.87,  0.00) ( -2.10, -0.16)
                                                 (  0.50,  0.00) :End matrix A
```

9.3 Program Results

```
 F08JVF Example Program Results

 Eigenvalues
     -7.0042           -4.0038            0.5968            3.0012

 Eigenvectors
                     1                 2                 3                 4
 1  ( 0.7293, 0.0000) (-0.2128, 0.1511) (-0.3354,-0.1604) (-0.5114,-0.0163)
 2  (-0.1654,-0.2046) ( 0.7316, 0.0000) (-0.2804,-0.3413) (-0.2374,-0.3796)
 3  ( 0.6081, 0.0301) ( 0.3910,-0.3843) (-0.0144, 0.1532) ( 0.5523, 0.0000)
 4  ( 0.1653,-0.0303) ( 0.2775,-0.1378) ( 0.8019, 0.0000) (-0.4517, 0.1693)
```